制御情報システム工学科　　制御工学実験　　指導書

動的システムの周波数特性解析（周波数応答法）:Scilab 版

１．目的

高次の動的システムの解析方法の一つである周波数応答法を理解する。周波数応答法は、現代制御理論やロバスト制御理論においても根幹となるシステム特性解析の概念であり、極めて重要である。本実験では、特に動的システムの周波数応答の表現法であるベクトル軌跡（ナイキスト線図）とボード線図の利用方法を修得することを目的とする。

２．理論概要

2.1　周波数伝達関数

伝達関数$G(s)$が次式で与えられる安定系：

$$G(s) = \frac{b_m s^m + b_{m-1} s^{m-1} + \cdots b_1 s + b_0}{(s-p_1)(s-p_2)\ldots(s-p_n)} \quad \cdots (2.1)$$

に対して，正弦波信号

$$u(t) = A\sin\omega t \quad \cdots (2.2)$$

を入力したときの出力 y(t)は

$$y(t) = A|G(j\omega)|\,sin(\omega t + argG(j\omega)) \quad \cdots (2.3)$$

となる。ここに、$G(j\omega)$は伝達関数$G(s)$に s=jω を代入したものであり、**周波数伝達関数**（Frequency Transfer Function）と呼ばれる複素関数である。その実数部を$a(\omega)$、虚数部を$b(\omega)$とすれば$G(j\omega)$は図１に示すように複素平面上にプロットした複素ベクトル$\vec{G}$で表せる。この複素ベクトルの絶対値 $|G(j\omega)|$ および偏角 arg$G(j\omega)$ は、$a(\omega)$，$b(\omega)$を用いて

$$|G(j\omega)| = \sqrt{a(\omega)^2 + b(\omega)^2} \quad \cdots (2.4)$$

$$\mathrm{arg}G(j\omega) = \tan^{-1}\frac{b(\omega)}{a(\omega)} = \phi(\omega) \quad \cdots (2.5)$$

と求めることができ、これらの絶対値と偏角を用いた指数関数表示によれば、周波数伝達関数は

$$G(j\omega) = |G(j\omega)|e^{j\phi(\omega)} \quad \cdots (2.6)$$

と表せる。(2.3) 式に示すように $|G(j\omega)|$ は、入出力信号の振幅比を与えるので、周波数ωにおける伝達関数の**ゲイン**と呼ばれる。また、arg$G(j\omega)$ は入出力信号の位相差を与えるので周波数ωにおける伝達関数の**位相**と呼ばれる。



図１　周波数伝達関数の複素平面での表現

2.2　ベクトル軌跡

図１に示したように周波数伝達関数は複素平面上にベクトル$\vec{G}$で表現される。周波数ωを 0 から+∞まで変化させるとベクトルの先端は軌跡を描く。この軌跡を**ベクトル軌跡**（vector locus）という。また、周波数ωを-∞から+∞まで変化させたときの軌跡を**ナイキスト**（Nyquist）**軌跡**と呼ぶ（ナイキスト線図とも呼ばれる。図２参照。）これらの軌跡の表示に際しては、軌跡には始点（ベクトル軌跡の場合はω=0 のとき）や終点（同じく、ω=+∞のとき）の他、虚軸や実軸と交差するといった特徴点をそのときのωの値と共に示すとともに、ω=0 から＋∞に向かって軌跡が変化する方向を矢印により示す。



図 2　ナイキスト線図



図 3　ボード線図

2.3　ボード線図

ベクトル軌跡が周波数応答を周波数伝達関数$G(j\omega)$の複素ベクトルの複素平面上に描く軌跡として一つの図により表示したのに対して、**ボード線図**（Bode diagram）は、ゲインと位相とをそれぞれに片対数（横軸に周波数ωを対数目盛でとる）プロットしたものである（図 3）。ゲインについては、デシベル値 $20\log_{10}|G(j\omega)|$ により表した**ゲイン曲線**（図 3 上図）で表し、位相については、度[deg]（またはラジアン[rad]）を単位にとって表した**位相曲線**（図 3 下図）で表示する。

ボード線図では、伝達関数 G(s)が複数（k 個）の伝達関数の積で表される場合には

$$20\log_{10}|G(j\omega)| = \sum_{i=1}^{k} 20\log_{10}|G_i(j\omega)| \quad \cdots (2.7)$$

$$argG(j\omega) = \sum_{i=1}^{k} argG_i(j\omega) \quad \cdots (2.8)$$

が成り立つ。(例えば文献[1] p.108)

このことから、G(s)のゲインおよび位相は、$G_i(s)$ のゲインおよび位相の和としてそれぞれ表されることとなる。高次の伝達関数の周波数特性を低次の基本的な伝達関数要素のボード線図から得られることは、ボード線図の特徴の一つである。

３．実験方法

3.1　周波数応答

3.1.1　一次遅れ系の正弦波応答

動的システム

$$Y(s) = P(s)U(s),\quad P(s) = \frac{1}{s+1}$$

において、入力を

$$u(t)=A\sin\omega t \quad (A > 0, \omega > 0)$$

としたとき、出力 y(t)がどのようになるか調べる。

(1)　**準備１**：ラプラス変換を用いると y(t)が次式となる。(実験前に次式を導出しておこう。レポートには必ず導出過程を記述すること。)

$$y(t) = \frac{A\omega}{1+\omega^2}e^{-t} + \frac{A}{1+\omega^2}(\sin\omega t - \omega\cos\omega t)$$

(2)　**準備２**：十分に時間が経過したとき、$lim_{t\to\infty}e^{-t} = 0$ が成り立つことから、定常状態での出力 $y_{ss}(t)$は次式のように近似できる。((1) と同様に実験前に導出しレポートに記載のこと。)

$$y_{ss}(t) = B(\omega)\sin(\omega t + \phi(\omega))$$

ただし、$B(\omega) = \frac{A}{\sqrt{1+\omega^2}},\quad \phi(\omega) = -\tan^{-1}\omega$

（3） A=1，ω＝0.5， 1，50 とした場合の入出力の波形 u(t),y(t)および y_{ss}(t)を、Scilab を用いて計算し, 時間応答を sce-file を作成してプロットする。(末尾付録 A.1 を参照のこと) そして、得られた波形振幅と位相について比較検討を行い、周波数の変化に伴うゲインと位相の変化について考察する。

3.1.2 二次遅れ系の正弦波応答

伝達関数 P(s)が $P(s) = \frac{1}{s^2+1.4s+1}$ と与えられる二次遅れ系について（3）と同様の A，ωの設定に対して入出力波形 u(t), y(t)（時間応答）を Scilab を用いて計算しプロットする。 そして、得られた波形振幅と位相の変化に着目して考察する。

3.2 ベクトル軌跡

二次遅れ系（標準形）で与えられる伝達関数$P(s) = \frac{K\omega_n^2}{s^2+2\zeta\omega_n s+\omega_n^2}$ のパラメータ K, ζ, ω_n（それぞれ、ゲイン定数、減衰係数、自然（固有）角周波数という）の変化に対するベクトル軌跡の変化を調べる。

（1） K=1, $\omega_n = 1$ と固定し，$\zeta = 0.25$, 0.5, 1 と変化させたときのナイキスト軌跡をプロットする。(付録 A.2 を参照）ベクトル軌跡の始点(ω=0)と終点（ω=＋∞）を求め、出力したナイキスト線図に追記すること。また、ω＝０～＋∞の範囲に対して描かれる部分を確認すること。(ヒント：この周波数範囲に対する$a(\omega)$，$b(\omega)$の符号を調べるとよい。)

（2） 3.1.2 の実験で作成した Scilab プログラムを用いて、減衰係数ζ の各値に対して振幅が１、角周波数をω＝0.1, 0.5, 1 の異なる正弦波入力信号を与えたときの正弦波応答 (定常応答波形）の変化を確認し、(1)で得たベクトル軌跡とその変化に整合することを確認する。なお、結果は図４に示すように整理すること。(この計算は、PC 性能によっては時間を要する。)

注）さらに興味があれば、付録 A０を実施する。



図４ 実験結果の整理例

3.3　ボード線図

3.3.1　ボード線図の作成

(1)　実験 3.2(1)と同様の設定（K=1, $\omega_n = 1$ と固定し，$\zeta = 0.25$, 0.5, 1 と変化させる）としたときの P(s)の周波数応答特性を表すボード線図を角周波数$\omega = 0.01$ から 100 と指定してプロットする。（付録 A.3 を参照）

(2)　設定した周波数域内でゲインが最大となる周波数（ピーク周波数）ω_p を求め，そのときのゲインの値（共振ピーク）M_p を Scilab により算出し記録する。

(3)　実験 3.2(2)の結果と比較してボード線図の特徴との対応を考察する。

3.3.2　2 つの伝達関数要素の直列結合とボード線図

(1)　伝達関数 $P(s) = \frac{s+0.1}{s+1}$ のボード線図を描く。

(2)　$P_1(s) = \frac{1}{s+1}$, $P_2(s) = s + 0.1$ のボード線図をそれぞれ描く。

(3)　ボード線図上で $P_1(s)$と $P_2(s)$のゲイン曲線と位相曲線とを加え合わせると $P(s)$のそれと一致することを確認する。

（$P(s)$, $P_1(s)$, $P_2(s)$　のゲイン曲線、位相曲線を一つの図にプロットして検討すること。）

4．研究課題

4.1　ゲイン位相線図について調べなさい。

4.2　一次遅れ系のナイキスト軌跡が円軌道を描くことを示しなさい。

5．参考文献

［1］　斉藤制海，徐粒：計測と制御シリーズ　制御工学　‐フィードバック制御の考え方ー，p.91-141, 森北出版（2011）

［2］　鈴木隆，板宮敬悦：例題で学ぶ自動制御の基礎，p.53-90，森北出版(2011)

［3］　中野道雄，高田和之，早川恭弘：機械工学入門講座　自動制御【第２版】, p.52-90，森北出版(2007)

［4］　野波健蔵，西村秀和：MATLAB による制御理論の基礎，p.97-142, 東京電機大学出版局(1998)

［5］　川田昌克，西岡勝弘：MATLAB/Simulink によるわかりやすい制御工学，p.108-132，森北出版(2005)

［6］　森泰親：演習で学ぶ基礎制御工学,p48-80，森北出版(2004)

付録

A.0 ベクトル軌跡（追加）

３．２（３）に続けて下記（１）、（２）を実施することにより、周波数応答の変化が理解できる.

（1）　さらに、K=1, $\zeta = 0.5$ と固定し、$\omega_n = 0.5,\ 1,\ 2$ と変化させせたときのベクトル軌跡をプロットするとともに、正弦波入力信号を(2)と同様に変化させたときの正弦波応答波形をプロットし、得られた結果を図４のように整理する。

（2）　つぎに、$\zeta = 0.5$, $\omega_n = 1$ と固定し、 K=0.5, 1, 5 と変化させせたときのベクトル軌跡をプロットし、(2)（3）の実験と同様に周波数を変化させた正弦波応答波形を求める。

実験3.2（2）		ζ		
		0.25	0.5	1
ω	0.1			
		ゲイン: 位相:	ゲイン: 位相:	ゲイン: 位相:
	0.5			
		ゲイン: 位相:	ゲイン: 位相:	ゲイン: 位相:
	1			
		ゲイン: 位相:	ゲイン: 位相:	ゲイン: 位相:

図４　実験結果の整理例

A.1　Scilab コマンド（sce-file）による正弦波入力応答シミュレーション

$P(s)=\frac{1}{s+1}$　なる系に入力を u(t)=10sin0.5t としたときの y(t) と定常状態での出力：

$y_{ss}(t)=|P(0.5j)|\sin\big(0.5t+\phi(0.5j)\big)$ の波形グラフを描く sci-file を参考として以下に示しておく。

リスト１：　動的システムの正弦波入力に対する応答プロット

```
s=%s;                              // 多項式演算子定義
P=syslin('c',1/(s+1))              // 伝達関数定義⇒連続システム変換
t=0:0.01:20;                       // 計算時間、サンプリング時間設定
A=10.0;                            // 振幅
Omeg=0.5;                          // 振動数
u=A*sin(Omeg*t);                   // 正弦波入力
y=csim(u,t,P);                     // 伝達関数出力演算
yss=A/(1+Omeg*Omeg)*(sin(Omeg*t)-Omeg*cos(Omeg*t)); // 収束理論式
plot2d(t,u,style=color(125,125,125));   // ２次元グラフ作成：時間,表示データ,灰色
plot2d(t,y,style=3);                    // ２次元グラフに追加
plot2d(t,yss,style=2);                  // ２次元グラフに追加
xtitle('Frequancy Response','time[s]','u, y, yss') // グラフ縦軸横軸凡例
xgrid();                                // グラフの補助線を描画
```



図 A.1　線形システムの応答波形（正弦波入力）

A.2　ナイキスト軌跡（ベクトル軌跡）

コマンド nyquist によりナイキスト軌跡(図 A.2)を描くことができる。以下に Scilab のコマンド例を示す。

```
>>s=%s;                          // 多項式演算子定義
>>P1=syslin('c',1/(s+1))         // 伝達関数定義
>>P2=syslin('c',1/(s^2+s+1))     // 伝達関数定義
>>nyquist([P1;P2],%t);           // ナイキスト線図表示
```



図 A. 2　ナイキスト線図

A.3　ボード線図

ボード線図の作成には関数 ”bode”を用いる。複数の伝達関数のボード線図を重ねて描くことも可能である。また、"freson"を用いて、最大ゲインでの周波数およびゲイン最大値を計算することができる。実行結果および作成されたボード線図を次頁に示す。

リスト２：　ボード線図の作成例１

```
s=%s;                                        // 多項式演算子定義
Omeg = 0.5;                                  // ω 値
Zeta = 0.1;                                  // ζ 値
G = Omeg*Omeg/(s^2+2*Zeta*Omeg*s+Omeg*Omeg);// 伝達関数定義
P = syslin('c',G);                           // 連続系変換
xset('window',0);                            // ウインドウ作成
bode(P);                                     // ω=0.01～100Hz のボード線図表示
T=3.0;
G1 = Omeg/(T*s+1);                           // 伝達関数定義
P1 = syslin('c',G1);                         // 連続系変換
xset('window',1);                            // ウインドウ作成
bode([P;P1]);                                // 複数ボード線図表示
Peak_freq = freson(P)                        // 伝達関数 G のゲインが最大値となる周波数[Hz]計算
w_freq = 2*%pi*Peak_freq                     // 角周波数 ω[rad/s]に変換
Max_G = 20*log10(abs(horner(G,%i*w_freq)))   // ゲイン最大値 dB
```

実行結果：

```
Peak_freq  =
   0.0787777
 w_freq  =
   0.4949747
 Max_G  =
   14.023048

 Execution done.
-->
```



図 A.3　″bode″関数によるボード線図の表示



図 A.4 ″bode″関数によるボード線図の重複表示

・phasemag 関数を用いたボード線図の作成

リスト３：　ボード線図の作成例２

```
s=%s;                                    // 多項式演算子定義
Omeg = 0.5;                              // ω 値
Zeta = 0.1;                              // ζ 値
G = Omeg*Omeg/(s^2+2*Zeta*Omeg*s+Omeg*Omeg);// 伝達関数定義
w = logspace(-2,2,400);                  // 対数表示
Gjw = horner(G,%i*w);                    // 周波数伝達関数設定
[Phase,GaindB] = phasemag(Gjw,'c');   // 位相<deg>,ゲイン<dB>の計算
xset('window',0);                        // ウインドウ作成
subplot(211);                            // ウインドウ分割２行１列の１行へ表示
plot2d(w,GaindB,logflag='ln',style=2);   // ゲイン(dB)グラフ表示
xtitle('Bode Diagram','ω[rad/s]','Gain[dB]')   // グラフ縦軸横軸凡例
xgrid();
subplot(212);                            // ウインドウ分割２行１列の２行へ表示
plot2d(w,Phase,logflag='ln',style=2);    // 位相グラフ表示
xtitle('','ω[rad/s]','Phase[degree]') // グラフ縦軸横軸凡例
xgrid();
```



図 A.5 phasemag 関数によるボード線図の描画